NOAA Technical Memorandum NESDIS AISC 10

A SECTORIZED STRETCHED GRIDMESH FOR MODELING SAN FRANCISCO BAY AND SHELF CIRCULATION

Kurt W. Hess

Washington, D.C.
April 1987

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Environmental Satellite, Data, and Information Service

NOAA TECHNICAL MEMORANDUMS

National Environmental Satellite Data and Information Services, Assessment and Information Services Center Subseries

The Assessment and Information Services Center provides value-added environmental information to support national needs and goals. The products and services of AISC's staff of environmental scientists, economists, librarians and information specialists, and computer specialists provide decision assistance to the nation's resource managers. AISC's programs in environmental assessments and environmental information management complement the data collection and archival program of other National Environmental Satellite Data and Information Services (NESDIS) components.

The Technical Memorandum series provides an informal medium for the documentation and quick dissemination of results not appropriate, or not yet ready, for formal publication in the standard journals. The series is used to report on work in progress, to describe technical procedures and practices, or to report to a limited audience.

These papers are available from the originating office and also from the National Technical Information Service (NTIS), U.S. Department of Commerce, 5285 Port Royal Road, Springfield, VA 22151. Prices vary for paper copy and microfiche.

NESDIS Technical Memorandum

AISC 1 Assessment Models for Surface Dispersion of Marine Pollutants. Kurt W. Hess, Fred G. Everdale, and Peter L. Grose, May 1985.

AISC 2 Comparison of Boundary Layer Winds from NWS LFM Output and Instrumented Buoys. Robert W. Reeves, and Peter J. Pytlowany, May 1985.

AISC 3 Assessments Model for Estuarine Circulation and Salinity, Kurt W. Hess, June 1985.

AISC 4 Development of a Tropical Cyclone Damage Assessment Methodology. Isobel C. Sheifer and John O. Ellis, January 1986.

AISC 5 Interdisciplinary Synoptic Assessment of Chesapeake Bay and the Adjacent Shelf. Edited by David F. Johnson, Kurt W. Hess, and Peter J. Pytlowany, August 1986.

AISC 6 Numerical Model of Circulation in Chesapeake Bay and the Continental Shelf. Kurt W. Hess, November 1986.

AISC 7 Application of AVHRR Satellite Data to the Study of Sediment and Chlorophyll in Turbid Coastal Water. Richard P. Stumpf, March 1987.

AISC 8 A Model for the Simulation of Larval Drift. David F. Johnson, March 1987.

AISC 9 Selective Retrieval of Spectral MRF Model Winds For Marine Applications. Peter J. Pytlowany, March 1987.

NOAA Technical Memorandum NESDIS AISC 10

A SECTORIZED STRETCHED GRIDMESH FOR MODELING SAN FRANCISCO BAY AND SHELF CIRCULATION

Kurt W. Hess

Marine Environmental Assessment Division
Assessment and Information Services Center

Washington, D.C.
April 1987

UNITED STATES DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary

National Oceanic and Atmospheric Administration
Anthony J. Calio, Undersecretary

National Environmental Satellite, Data, and Information Service
Thomas N. Pyke, Jr.
Assistant Administrator

CONTENTS

A SECTORIZED STRETCHED GRIDMESH FOR MODELING SAN FRANCISCO BAY AND SHELF CIRCULATION

Kurt W. Hess

Marine Environmental Assessment Division
Assessment and Information Services Center
National Environmental Satellite, Data, and
Information Service, Washington, DC 20235

ABSTRACT. A numerical circulation model gridmesh with stretched coordinates is developed and applied to San Francisco Bay and the adjacent shelf seaward to 50 km from Pt. Arena in the north to Pt. Sur in the south. The new mesh has variable grid widths, which are created by transforming (stretching) the x and y coordinates. The transformation of either axis can be applied in two sectors so that grid cells inside the Bay are small, and cells along the outer coast are larger. The transformed mesh is very useful because (1) it provides high resolution inside the Bay where it is needed, and (2) the larger coastal grid cells will significantly reduce computer running time.

1. INTRODUCTION

The Marine Environmental Assessment Division has continued development and testing of the general three-dimensional free-surface numerical circulation model MECCA, (Model for Estuarine and Coastal Circulation Assessment), originally described by Hess (1985a). Models such as this have become important tools for studying the physical dynamics of estuaries and exploring the consequences of various human uses of the resource. MECCA uses finite difference approximations to the momentum, mass, continuity, and concentration equations to simulate three-dimensional water currents and salinities at 10 levels in a shallow water domain at time scales ranging from a few minutes to several months, and space scales ranging from a few kilometers to a few hundred kilometers. A user's guide (Hess, 1985b) is available for the program.

The model was first applied to Chesapeake Bay to simulate tidal and density-driven currents, and to study the reduction in salinity in the Bay during the high river discharge conditions following the passage of hurricane Agnes over the watershed in

1972 (Hess, 1985a). MECCA was subsequently applied to a larger domain which includes Chesapeake Bay and a portion of the local continental shelf (Hess, 1986). Circulation in the Bay-shelf region is important for assessing the biological productivity of the estuary. Wind forecasts and satellite-derived temperatures have been used to provide input to the circulation model, which generated currents that in turn were used to simulate trajectories of hypothetical biological drifters and blue crab larvae (Johnson et al., 1986; 1987).

Experience in modeling the Chesapeake Bay-shelf region has shown that finer gridmesh resolution is highly desirable for the part of the model domain covering Chesapeake Bay, since the land and channel features are on the order of only a few kilometers. On the continental shelf, the grids can be much larger, since we are not interested in as much detail. The most efficient way to achieve this variation in grid resolution is by the use of a stretched coordinate system (Sheng, 1983).

The San Francisco Bay and the adjacent coastal region has been selected as the first region of study in a proposed simulation of the transport and fate of the larval stages of Dungeness crab. It is expected that some of the techniques developed in Johnson et al. (1986, 1987) for the blue crab can be directly applied to this west coast crab fishery.

2. BACKGROUND DATA AND MODELS FOR SAN FRANCISCO BAY

Since San Francisco Bay is located within a rapidly-growing and environmentally-conscious area, there is a wealth of oceanographic data on the estuary. Descriptions of San Francisco Bay environmental studies are given in two overviews of the areas, one edited by Conomos (1979) and another edited by Cloern and Nichols (1985). A unique summary of the impacts of oceanic and weather events on the region during 1985 is presented by Dowgiallo et al., 1986.

The physical oceanography of the area has also been well-studied. Analyses of the tides in the Bay are given by Walters (1982), Walters and Gartner (1985), and Cheng and Gartner (1985). A discussion of the non-tidal current null zone is given in Peterson et al. (1975). A review of the NOAA circulation survey of the Bay is given in Welch et al. (1985). Background information on the west coast shelf can be found in the preliminary published results of the Coastal Ocean Dynamics Experiment (CODE) (American Geophysical Union, 1987).

Several numerical modeling studies of San Francisco Bay have been carried out by R. Cheng and his associates at the U.S. Geological Survey in Menlo Park, California. A study of the two-dimensional (horizontal) Lagrangian residual circulation in South San Francisco Bay using finite difference techniques was presented in Cheng and Casulli (1982). A model of the horizontal tidal and residual circulation in the whole Bay was reported by Cheng and Walters (1982), who used a finite element approach. A two-dimensional modeling study of currents and salinity distribution in Suisun Bay is given in Smith and Cheng (1985). A discussion of Eulerian-Lagrangian techniques for solving the convection diffusion equation is given in Cheng et al. (1984).

The present modeling study differs from the above studies in two ways: (1) it is designed to simulate flow in the Bay, Gulf of the Farralones and beyond, rather than to study just the circulation inside the Bay proper, and (2) it employs a three-dimensional numerical circulation model.

3. GRID COORDINATE STRETCHING

Variable grid sizes can be obtained by stretching a square-celled grid in one or more directions by the use of a coordinate transformation. Axis transformation to produce variable-width cells is preferable to the direct use of unequally spaced points because the latter will reduce the accuracy of spacially centered finite-difference approximations from second order to first order (Roach, 1972).

In a uniform mesh with cells of size ΔL, the numerical solution is generated in modeled horizontal space (r,s) which is related to geographic space (x,y) by

$$x = r\Delta L \quad \text{and} \quad y = s\Delta L \qquad (3.1)$$

so that x and r have a simple linear relationship. By contrast, variable grid sizes in real space are generated by applying a nonlinear stretching transformation to get x, although the numerical calculations are still performed on a uniform mesh.

3.1 The MECCA Uniform Gridmesh

A brief explanation of MECCA's present, uniform numerical gridmesh is useful to introduce the discussion. The mesh consists of square cells with certain designated cells representing water. The gridmesh represents horizontal space (x,y) as a flat plane, tangent to the earth's surface. The tangent plane approximation is valid as long as the horizontal extent of the mesh covers a distance not more that a few percent of the earth's radius (1 percent is 63.71 km).



Figure 1. Representation of the model gridmesh, its orientation with respect to the earth's surface, and the latitude-longitude reference point.

The mesh is stored in a data file as a matrix whose rows (index M) increase downward representing the x direction, and whose columns (index N) increase to the right representing the y direction. The mesh is referenced to the earth's surface by specifying the latitude and longitude of the lowest corner of a specific cell, designated (M_c,N_c), as shown in Fig. 1. The subscript c refers to the Coriolis parameter, which is computed from the latitude of this point. The lowest corner of a cell is defined as the one closest to the origin of the x,y (or M,N) coordinate system. A rotation angle, θ, indicates how much clockwise rotation of the mesh is needed to align it with the geographic coast.

The numerical calculations for the transformed system are performed on a gridmesh essentially identical to the one described above. The influence of the axis transformations appears in coefficients multiplying the finite-difference equivalents of the horizontal derivatives, as explaned in the next section.

3.2 The Transformation Function

The generalized transformation we use has been successfully applied to a model of Mobile Bay and Sound (Sheng, 1983; Schmalz, 1985), and is

$$x = a + br^c \tag{3.2}$$

where x is the coordinate in real space, and r is the coordinate in the uniform (model) domain. There is an analogous transformation between the y and s coordinates. In the following equations, partial differentiation will be indicated when the dependent variable is followed by a subscripted comma and the independent variable. Therefore, using the transformation (3.2), we get

$$f_{,x} = [cbx^{c-1}]^{-1} f_{,r} = e_x f_{,r} \qquad (3.3)$$

The general strategy for applying the transformation to the system of equations is to substitute the expression (3.3) for the x-derivative wherever it appears in the hydrodynamic equations. The y derivative is treated analogously.

Consider the transformation applied to a single zone of the x axis with decreasing grid sizes. Along the r axis there are n equally-spaced intervals corresponding to integral values $r = r_1$, r_2, . . . r_m. Note that $n = r_m - r_1$. Along the x axis there is an equal number of unequally-spaced intervals between points X and X', as shown in Fig. 2.

For this zone there are three unknown parameters: a, b, and c. Thus three conditions are needed to uniquely determine the stretching function. Two obvious ones are found by applying (3.2) to the end points. However, there is no obvious third condition. More information is needed to complete the solution.

The approach taken here is based on the user's selection of a maximum gridcell size, D_{max}, and a minimum gridcell size, D_{min}, to apply throughout the zone. This gives some indication of how many cells will be needed to cover the interval. For this example, with decreasing cell size, the limiting size for the first cell, D_F, is the maximum gridcell size, and the limiting size for the last cell, D_L, is the minimum gridcell size, i.e.

$$D_F = D_{max} \qquad D_L = D_{min} \qquad (3.4)$$

```
            <-----(stretching zone)------>

                 X                              X'
+--------+----------+-------+-----+----+-----> x
0        x1         x2            xm-1 xm

+------+------+------+------+------+--------> r
0      r1     r2            rm-1   rm
```

Figure 2. Schematic of a simple, single-zone coordinate transformation between the x and r axes with decreasing grid sizes. Here any point x_i is related to r_i by an equation like (3.2).

Then, for the largest interval in our example (Fig. 2), the following must hold:

$$x_2 - x_1 \leq D_F \tag{3.5}$$

and for the smallest interval:

$$x_m - x_{m-1} \geq D_L \tag{3.6}$$

It can be shown that (3.5) is met if

$$x_{,r} = D_F \quad \text{at } r = r_1 \tag{3.7a}$$

and (3.6) is met if

$$x_{,r} = D_L \quad \text{at } r = r_m \tag{3.7b}$$

These become two of our required conditions. The third condition needed to determine the stretching function was mentioned above, and is simply the application of (3.2) to the initial point

$$x = X \text{ at } r = r_1 \tag{3.7c}$$

Since three conditions have been specified, the second end condition, $x = X'$ at $r = r_m$, can only be met approximately, since r_m is an integer. The actual value of the end point produced by the transformation is

$$X_e = [a + b{r_m}^c] \tag{3.8}$$

The method of solution for a, b, and c thus requires testing values of n in the range

$$2 \leq n \leq 2(X' - X)/(D_{max}D_{min})^{1/2} \tag{3.9}$$

and selecting $r_m = r_1 + n$ so that

$$E = \left| X' - X_e \right| \tag{3.10}$$

is a minimum. By varying X, D_{max}, and D_{min}, the value of E can usually be made as small as desired. The three unknowns are then evaluated as

$$c = 1 + \ln(D_F/D_L)/\ln(r_m/r_1) \tag{3.11a}$$

$$b = D_F/(c{r_1}^{c-1}) \tag{3.11b}$$

$$a = X - b{r_1}^c \tag{3.11c}$$

The values of a, b, and c can be uniquely determined by (3.11) unless $r_1 = 0$, which occurs in the first zone. This problem is easily overcome by making the first zone one of uniform cell sizes (see Sections 3.3 and 3.4).

3.3 Transformation for Increasing or Uniform Widths

The above discussion has assumed that the grid size was decreasing in the x direction, but there can also be increasing or uniform width zones. For increasing grid sizes,

$$D_F = D_{min} \qquad D_L = D_{max} \tag{3.12}$$

and (3.7a,b,c) and (3.11a,b,c) still hold.

It should be noted that in general it is not possible to generate a transformation giving a sequence of widths symmetrical around an arbitrary point in the x axis. Normally, however, approximate symmetry will hold if the widths of the zones are equal.

A special case arises when the grid widths are increasing and $D_{max}/D_{min} \geq 5$. To force the widths to be more nearly symmetrical to those with decreasing grid width, the zone is subdivided into two new zones; the border between the two is arbitrarily defined to be at $x = X + 0.4(X' - X)$, and the slope condition at the boundary is forced to be $x_{,r} = 2D_F$.

For the case of uniform widths, the grid cell size D (either D_{max} or D_{min}) is first selected. Then the number of cells is

$$n = \text{maximum of } \{\ 1\ ,\ I\{(X' - X)/D\}\} \tag{3.13}$$

where I{} is the integer closest to the value of the argument. For this case,

$$c = 1 \tag{3.14a}$$

$$b = D \tag{3.14b}$$

$$a = X - br_1 \tag{3.14c}$$

3.4 Multizone Transformation

The stretching transformation along one axis is more useful when applied in a piecewise manner to allow, for example, a region of decreasing grid spacing to precede a region with increasing grid spacing. In this case, the axis is divided into zones denoted by the subscript i, in each of which the transformation is

$$x_i = a_i + b_i r^{c_i} \qquad (3.15)$$

If i = 1, the values of a, b, and c can be selected as in the preceding section. Otherwise, there is a requirement for continuity of x and its first derivative with respect to r at the boundary between zones. Continuity of the function is satisfied if X_e from the previous zone is taken for X. The derivative condition is automatically satisfied by (3.7a,b), since

$$D_{L_i} = D_{F_{i+1}} \qquad (3.16)$$

3.5 Sectorizing

In general, the transformation along (for example) the x axis holds for all values of y in the domain. However, this need not be so. We will now develop the principles of applying one multizone transformation for a specified range of y values, called a sector, and another transformation to another sector of y values.

Consider a hypothetical estuary separated from the open coast by a narrow mouth, as shown in Fig. 3. The line $y = y^*$ has been drawn parallel to the outer coastline through the mouth, and separates the modeled region (the area within $0 \leq x \leq x_o$ and $0 \leq y \leq y_o$) into two sectors. The sector closest to the origin is called Sector I and the other is called Sector II. In Sector I of this hypothetical estuary the grid cell widths in the y direction are uniform. But in Sector II there are three zones of varying cell widths: decreasing, uniform, and increasing. The position of the zones has been carefully chosen so that the uniform grids in each sector match at the bay mouth.

The model grid resulting from the transformation will be computationally advantageous because it will have a relatively small number of grids covering the coastal area in comparison with the number inside the estuary.

In the numerical scheme, which assigns coefficients e_x in Sector I, and e'_x in Sector II, there will be no computational problems in the x-direction flow because any given row will be in either one sector or the other. For y-direction flow, there will be no problems as long as at cells other than those at the mouth there is either a barrier or at least one row of land cells separating the sectors (to ensure zero velocity). Thus the necessity of an estuary physically separated from the coast is apparent.



Figure 3. Hypothetical estuary connected to the open coast by a narrow mouth. The modeled region is to be contained within the space $0 \leq x \leq x_o$ and $0 \leq y \leq y_o$. The line $y = y^*$ crosses the mouth and separates the sectors. Above the line in Sector II, an x-direction stretching transformation produces three zones of grid cells of varying widths. In the first zone, the widths are decreasing in the x direction; in the second zone, the widths are uniform; and in the third zone, the widths are increasing. Below the sector dividing line in Sector I, there is a single zone with uniform cell widths. Transformation in the y direction (not shown) is independent of the x-direction transformation.

4. APPLICATION TO SAN FRANCISCO BAY AND SHELF

Consider the configuration of San Francisco Bay and the local shelf from Pt. Arena on the north to Pt. Sur in the south (Fig.4). The small area of the Bay and the vastness of the coastal section plus the existance of the narrow mouth make the area ideal for sectorized stretching.

4.1 Sectorizing Lines

The first step in the sectorizing process is to choose a sector dividing line; in the case of San Francisco Bay one was selected which runs across the Golden Gate along a line oriented 32.0° west of north (Fig. 4). This line is denoted A-A and was selected because (1) it is approximately parallel to the coast, (2) it crosses the mouth near its narrowest point, (3) all of San Francisco Bay is landward, and (4) all of the coastal area, including Monterey Bay, is seaward. The modeled domain can then be chosen to enclose the coastal section bounded by Pt. Arena on the north and Pt. Sur on the south. The object is to find a sectorized transformation so that the Bay, in Sector I, has uniformly small grid sizes (of roughly the width of the entrance) for better resolution, and the coastal region, in Sector II, has grid widths varying along the coast so that a large portion of the coast can be covered.

After selecting the sector dividing line, we choose a line (B-B) for the northern boundary of the outer coastal region which (1) is normal to the sector dividing line A-A and (2) passes just north of Pt. Arena. The second normal line, C-C, passes through the coast just north of the Bay mouth. This is the common line, and will be the same distance from B-B in each sector. The third normal line, D-D, crosses the coast just south of Pt. Sur to form the southern boundary. The distance from the northern boundary (B-B) to the common line (C-C) is approximately 90 nautical miles (nmi), and the distance from B-B to the southern boundary just below Pt. Sur (D-D) is approximately 194 nmi (Fig. 4). The exact position of these lines can be adjusted to locate the common line more precisely.

We need to establish northern and southern boundaries for the Bay itself. With the chosen orientation, the distance from the northern boundary (B-B) to the northern portion of San Pablo Bay (E-E) is about 74 nmi, and from the northern portion of San Pablo Bay to the southern section of south San Francisco Bay (F-F) is 46 nmi. The landward extent of the Bay riverine system is taken to be eastward of the sector dividing line by about 40 nmi, at line G-G. The coastal boundary (H-H) can be selected to suit the modeler's purpose. We take it to lie 80 nmi west of A-A.



Figure 4. Base map of San Francisco Bay and the local shelf, showing the position of the sector dividing line (A-A), the northern mesh boundary (B-B), the common line (C-C), and the southern mesh boundary line (D-D). The eastern boundary line is G-G, and the Bay is bounded by E-E and F-F. The coastal boundary is at H-H.

4.2 Generation of the Gridmesh

The next step in generating the sectorized mesh is the selection of maximum and minimum grid cell sizes. Since our computer program, MECCA, presently has maximum array sizes of 40 by 60 grids, the outer coast width of 194 nmi can easily be covered by 20 cells with a 10 nmi maximum cell size. A minimum grid cell size of 1.5 nmi was chosen because it is roughly the width of the Golden Gate, and the Bay can be covered by 31 cells of this size in the north-south direction.

The transformation functions are generated by a computer program (Appendix) using the methods described in Chapter 3. We start with the y axis transformation. The cell sizes D_{max} and D_{min} are required input.

The input data for each zone consists of the selection of

1. the type of variation (increasing, uniform, or decreasing), and
2. the target axis value for the end of the zone (Y').

For the first zone, the value of s corresponding to Y is also required, and if the zone has uniform grid sizes, the size (D_{max} or D_{min}) must be selected. The output for each zone is the actual end point [Y_e], the grid coordinate for the end [s], and of course a, b, and c.

We generate the attributes of Sector II first, because it's more complex than Sector I. All distances are in nautical miles. For the first zone the set of input values is (uniform, Y'=60, D_{max}) with Y_a at s=0 and the results are [Y_e=60, s=6] (the generated values of y corresponding to boundaries between all grid cells in all sectors are shown in Table 1). The second zone, with the input (decreasing, Y'=90), gives the results [Y_e=91.476, s=14] for the distance from B-B to C-C. This is close to our estimate of 90 nmi, so we can select this line to be the common line, C-C; the line B-B is then relocated to be 91.476 nmi to the north.

Continuing, the next zone input is (uniform, Y'=93) and results in a single grid at the mouth with [Y_e=92.976, s=15]. The next zone (increasing, Y'=122) results in [Y_e=121.657, s=23] (actually two zones are needed to cover this interval). The last zone has input (uniform, Y'=194), giving [Y_e=191.657, s=30]. The values of a, b, and c for all zones in this and the other sectors are given in Table 2.

Sector I is to consist entirely of uniform grid cells in the y direction. In order for the sectors to match, the common line C-C must have the same y-value in each sector. In Sector II, the line C-C is at y = 91.476 (s = 14). A uniform grid for Sector I

Table 1. Grid distances in nautical miles for the untransformed x and y axes. D is the grid size.

	Y-Sector II		Y-Sector I		X-Sector I	
n	y	D	y	D	x	D
0	0.000		70.476		0.000	
1	10.000	10.000	71.976	1.500	1.500	1.500
2	20.000	10.000	73.476	1.500	3.000	1.500
3	30.000	10.000	74.976	1.500	4.500	1.500
4	40.000	10.000	76.476	1.500	6.000	1.500
5	50.000	10.000	77.976	1.500	7.500	1.500
6	60.000	10.000	79.476	1.500	9.000	1.500
7	68.419	8.419	80.976	1.500	10.500	1.500
8	74.520	6.100	82.476	1.500	12.000	1.500
9	79.124	4.604	83.976	1.500	13.500	1.500
10	82.710	3.586	85.476	1.500	15.000	1.500
11	85.574	2.864	86.976	1.500	16.500	1.500
12	87.909	2.335	88.476	1.500	18.000	1.500
13	89.846	1.937	89.976	1.500	19.500	1.500
14	91.476	1.630	91.476	1.500	21.000	1.500
15	92.976	1.500	92.976	1.500	22.500	1.500
16	94.601	1.624	94.476	1.500	24.000	1.500
17	96.489	1.888	95.976	1.500	25.500	1.500
18	98.665	2.176	97.476	1.500	27.000	1.500
19	101.152	2.487	98.976	1.500	28.500	1.500
20	103.975	2.824	100.476	1.500	30.000	1.500
21	107.711	3.735	101.976	1.500	31.500	1.500
22	113.337	5.627	103.476	1.500	33.000	1.500
23	121.657	8.320	104.976	1.500	34.500	1.500
24	131.657	10.000	106.476	1.500	36.000	1.500
25	141.657	10.000	107.976	1.500	37.500	1.500
26	151.657	10.000	109.476	1.500	39.000	1.500
27	161.657	10.000	110.976	1.500	40.500	1.500
28	171.657	10.000	112.476	1.500	42.118	1.618
29	181.657	10.000	113.976	1.500	43.989	1.871
30	191.657	10.000	115.476	1.500	46.142	2.154
31			116.976	1.500	48.611	2.468
32			118.476	1.500	51.426	2.815
33			119.976	1.500	55.145	3.719
34			121.476	1.500	60.731	5.586
35			122.976	1.500	69.021	8.290
36					79.021	10.000
37					89.021	10.000
38					99.021	10.000
39					109.021	10.000
40					119.021	10.000

Table 2. Parameters used for the x and y direction transformations. Six place precision is needed to generate distances to the thousandth of nmi, as in Table 1.

X Direction

Sector	Zone	Length	a_i	b_i	c_i
I	1	27	0.000000E+00	0.150000E+01	0.100000E+01
	2	5	0.325272E+02	0.427197E-06	0.507976E+01
	3	3	0.447754E+02	0.124578E-20	0.144354E+02
	4	5	-.280979E+03	0.100000E+02	0.100000E+01

Y Direction

Sector	Zone	Length	a_i	b_i	c_i
I	1	35	0.704760E+02	0.150000E+01	0.100000E+01
II	1	6	0.000000E+00	0.100000E+02	0.100000E+01
	2	8	0.108425E+03	-.445882E+03	-.123902E+01
	3	1	0.704764E+02	0.150000E+01	0.100000E+01
	4	5	0.863770E+02	0.645226E-03	0.340942E+01
	5	3	0.977347E+02	0.193435E-11	0.961445E+01
	6	7	-.108343E+03	0.100000E+02	0.100000E+01

is chosen so that Y = 91.476 - 14(1.50) = 70.476 at s = 0. With a total of 33 cells, the southern extent (Y_e) is at 119.976 nmi, which is enough to cover the whole Bay.

The stretching for the x direction requires only one sector. We again chose 1.50 nmi as the minimum cell size and 10 nmi as the maximum. With this minimum, we set the eastern boundary (E-E) to be at X=0 at r=0. With the input (uniform, X'=40, D_{min}), A-A is at X_e=40.5 (r=27) to the west. In the next zone with input (increasing, X_b=70), the end is at [X_e=69.021, s=35]. The last zone of uniform cell sizes (uniform, x'=120) results in [X_e=119.021, s=40].

4.3 Latitude-Longitude Reference

As discussed in Section 3.1, the location and orientation of the mesh in reference to the earth's surface must be specified. After close inspection of the Golden Gate (Fig. 5), we position the line A-A (oriented 32^o west of north) to pass close to Pt. Bonita, and the grid column, N = 15, to be approximately centered



Figure 5. Close up of the gridmesh in the vicinity of the Golden Gate. The sector dividing line, A-A, is positioned close to Pt. Bonita, and the common line (C-C) is at the lower side of the grid column representing the Bay mouth (N=15). The latitude-longitude reference point is at the lower corner (i.e. closest to N-1, M-1) of grid cell (N=15, M=27), near Pt. Diablo.

on the Bay mouth. The mesh reference point (see Fig. 1) is then selected to be near Pt. Diablo at $N_c=15$, $M_c=27$, and latitude 37^o 49.415'N and longitude 122^o 29.8076'W. The whole mesh and the coastline can now be drawn (Fig. 6).

4.4 The Transformation Function in MECCA

Consider the x-direction transformation which consists of I zones, each with a length (number of cells) of L_j. The distance, x_1, to the lower side of any cell, M, in the grid mesh is given by the following

$$x_1 = a_i + b_i(M-1)^{c_i} \tag{4.1}$$

where i is such that

$$i = 1 \qquad \text{for} \quad M \leq L_1 \tag{4.2a}$$

$$\sum_{j=1}^{i-1} L_j < M \leq \sum_{j=1}^{i} L_j \qquad \text{for} \quad M > L_1 \tag{4.2b}$$

and I, L_j, a_i, b_i, and c_i depend upon the sector. Table 2 shows the values used to get the distances in Table 1.

In the mesh, all cells in rows

$$1 \leq M \leq M_{sector} \tag{4.3}$$

lie in Sector I. Other rows are in Sector II. Here, $M_{sector}=28$.

4.5 Selection of Water Grids

The water grids are selected by placing a scaled plot of all grids in the mesh over Mercator projections of the Bay and shelf area. The NOAA navigation charts used are Numbers 18010, 18640, and 18680. The resulting water grid mesh is shown in Fig. 7.

5. OTHER CONSEQUENCES OF TRANSFORMING

5.1 Averaging

Another consequence of the stretching transformation is that spacial averaging in the finite difference scheme becomes more complicated. Consider the function f(x) defined over the interval X_a to X_b shown in Fig. 1. Suppose that f varies linearly and that we need to estimate $f(x_2)$ from its values at x_1



Figure 6. The grid mesh plotted over the California coastline in the region of San Francisco Bay.



Figure 7. Water cells in the final gridmesh.

and x_3. Then

$$\bar{f}(x_2) = [(x_3 - x_2)f(x_1) + (x_2 - x_1)f(x_3)]/(x_3 - x_1) \quad (5.1)$$

or in the r coordinate system,

$$\bar{f}_2 = [(x_3-x_2)/(x_3-x_1)]f_1 + [(x_2-x_1)/(x_3-x_1)]f_3$$

$$= A_1f_1 + A_3f_3 \quad (5.2)$$

where A is the variable weighting factor. For uniform spacing, A = 0.5, but for a general gridmesh, A(x,y) must be evaluated everywhere.

5.2 Explicit Time Step

For a square-grid mesh, Sobey (1970) has shown that a typical explicit numerical scheme has, for the linearized, frictionless, constant-depth (h_o) case, a limiting timestep (Δt) of

$$\Delta t \leq \Delta x/(2gh_o)^{1/2} \quad (5.3)$$

where Δx is the cell width. By examining the algebra used to derive (3.14), we find that for unequal grid widths, Δx and Δy, the equivalent limiting timestep is

$$\Delta t \leq [\{\Delta x^{-2} + \Delta y^{-2}\}/2]^{-1/2}/(2gh_o)^{1/2} \quad (5.4)$$

Note that (5.4) reduces to (5.3) if $\Delta x = \Delta y$.

6. SUMMARY AND FUTURE PLANS

The above discussion shows that a grid transformation scheme can be applied successfully to a coastal region. The process involves much trial and error, but the outline of the procedure remains the same. The crucial steps are (1) selection of the sectorizing line, (2) selection of the boundary and common lines, (3) chosing the maximum and minimum grid sizes to give the best fit to the distances, and (4) aligning the grid with local features. These were completed successfully for San Francisco Bay.

MECCA has been recoded to accomodate variable cell sizes and is now being tested. One test will be to compare tides and currents with the older version (Version 3.0) to those generated with the new (Version 4.0), using the stretching transformation but with uniform cell sizes. Another test will be to compare tides and currents from Version 3.0 with Version 4.0 and variable grids.

Future applications involve generating the sectorized stretched mesh for Chesapeake Bay and the adjacent shelf. For San Francisco Bay, the bathymetry needs to be taken from the charts. Once the model is calibrated and verified, the model can be used to simulate conditions using historical buoy data for winds, currents, and water density. Comparisons can also be made with satellite-derived sea surface temperatures.

A longer-term project involves the study of Dungeness crab larval drift along the west coast between California, Oregon, and Washington. Circulation modeling may require an extended grid network, equations in spherical coordinates, and a sheared coordinate system.

7. REFERENCES

American Geophysical Union, 1987: Special Section: Coastal Ocean Dynamics Experiment (CODE), J. Geophys. Res., 92, 1455 - 1885.

Cheng, R. T., and V. Casulli, 1982: On Lagrangian residual currents with applications in South San Francisco Bay, California. Wat. Resour. Res., 18, 1652 - 1662.

Cheng, R. T., V. Casulli, and S. N. Milford, 1984: Eulerian-Lagrangian solution of the convective-dispersion equation in natural coordinates. Wat. Resour. Res., 20, 944 - 952.

Cheng, R. T., and J. W. Gartner, 1985: Harmonic analysis of tides and tidal currents in South San Francisco Bay, California. Est. Coast. Shelf Sci., 21, 57 - 74.

Cheng, R. T., and R. A. Walters, 1982: Modelling of estuarine hydrodynamics and field data requirements. Finite Elements in Fluids, 4, 89 - 108.

Cloern, J. E., and F. H. Nichols (eds.), 1985: Temporal Dynamics of an Estuary: San Francisco Bay. Junk Publishers, Boston, 237 pp.

Conomos, T. J. (ed.), 1979: San Francisco Bay - the Urbanized Estuary. Pacific Div., American Assn. Adv. Sci., San Francisco, Ca, 493 pp.

Dowgiallo, M. J., I. C. Sheifer, F. G. Everdale, M. C. Predoehl, K. B. Pechman, S. Z. Green, K. W. Hess, and R. P. Stumpf, 1986: Marine Environmental Assessment: San Francisco Bay, 1985 Annual Summary, 122 pp.

Hess, K. W., 1985a: Assessment Model for Estuarine Circulation and Salinity. NOAA Technical Memorandum NESDIS AISC 3, National Environmental Satellite, Data, and Information Service, NOAA, U.S. Department of Commerce, 39 pp.

___, 1985b: User's guide to running MECCA on the AISC VAX 11/750 and the DAMUS UNIVAC 1100. National Environmental Satellite, Data, and Information Service, NOAA, U.S. Department of Commerce, 40 pp.

___, 1986: Numerical Model of Circulation in Chesapeake Bay and the Continental Shelf. NOAA Technical Memorandum NESDIS AISC 6, National Environmental Satellite, Data, and Information Service, NOAA, U.S. Department of Commerce, 47 pp.

Johnson, D. F., K. W. Hess, and P. J. Pytlowany, 1986: Interdisciplinary synoptic assessment of Chesapeake Bay and the adjacent shelf. NOAA Technical Memorandum NESDIS AISC 5, National Environmental Satellite, Data, and Information Service, NOAA, U.S. Department of Commerce, 90 pp.

___, 1987: Circulation modeling as an aid to management of the blue crab fishery in Chesapeake Bay. Proceedings, Estuarine and Coastal Management: Tools of the Trade (in press).

Peterson, D. H., T. J. Conomos, W. W. Broenkow, and P. C. Doherty, 1975: Location of the non-tidal current null zone in northern San Francisco Bay. Est. Coast. Mar. Sci., 3, 1 - 11.

Roache, P. J., 1972: Computational Fluid Dynamics, Hermosa Publishers, Albuquerque NM, 434 pp.

Schmalz, R. A., 1985: Numerical model investigation of Mississippi Sound and adjacent areas. U. S. Army Corps of Engineers, Coastal Engineering Research Center Miscellaneous Paper CERC-85-2, 230 pp plus appendices.

Sheng, Y. P., 1983: Mathematical modeling of three-dimensional coastal currents and sediment dispersion: model development and application. U. S. Army Corps of Engineers, Coastal Engineering Research Center Tech. Report CERC-83-2, 288 pp.

Smith, L. H., and R. T. Cheng, 1985: Tidal and tidally-averaged circulation characteristics of Suisun Bay, California. (Submitted to Wat. Resour. Res.)

Sobey, R. J., 1970: Finite-difference schemes compared for wave-deformation characteristics in mathematical modeling of two-dimensional long-wave propagation. U.S. Army Corps of Engineers, Coastal Engineering Research Center Technical Memorandum 32, 29 pp.

Walters, R. A., 1982: Low-frequency variations in sea level and currents in South San Francisco Bay. J. Phys. Oceanog., 12, 658 - 668.

___, and J. W. Gartner, 1985: Subtidal sea level and current variations in the northern reach of San Francisco Bay. Est. Coast. Shelf Sci., 23, 17 - 321

Welch, J. M., J. W. Gartner, and S. K. Gill, 1985: San Francisco Bay area circulation survey: 1979-1980. National Ocean Service Oceanographic Circulation Survey Rept. 7, 180 pp.

APPENDIX. Computer Program Listing

The following is a listing of a computer program designed to calculate variable grid distances and the stretching parameters a, b, and c given the maximum and minimum grid sizes, and initial x distance, and intermediate zone boundary distances.

```
C         PROGRAM SGRIDX  (STRETCHED GRID)
C         JULY 18, 1986   K.W. HESS    MEAD    VAX 11/780
C         PURPOSE - TO GENERATE GRID LINES
C                   USING THE FORM:   X = A + B*R**C
C                   OR                R = ((X-A)/B)**(1/C)
C                   TO COMPILE:  F SGRID
C                                DI3LOAD SGRID T10
C         VARIABLES -
C              A1,B1,C1 = STRETCHING PARAMETERS
C             DMAX,DMIN = MAXIMUM, MINIMUM GRID WIDTHS (IN X SPACE)
C                         ALLOWED
C                    IT = ZONE TYPE (1=DECREASE, 2=UNIFORM, 3=INCREASE)
C                 IZONE = NUMBER OF PRESENT ZONE
C                 JCALL = NUMBER OF SETS OF INPUT VALUES (NO. OF ZONES
C                         MAY BE LARGER)
C                  MCUM = CUMULATIVE GRIDS TO THE PRESENT ZONE
C                    MS = NUMBER OF CELLS IN THE ZONE
C                    XT = TARGET DISTANCE FROM XO TO END OF ZONE
C                 XLAST = ACTUAL (COMPUTED) DISTANCE FROM XO TO END OF ZONE
C                    X1 = DISTANCE FROM XO TO START OF ZONE
C                    XO = X VALUE AT R=0
      COMMON/VALUES/XPLOT(200),JSEG(200),MMAX,KSEG(50),XL(361)
      COMMON/VALUE2/IZONE,IU,A1,B1,C1,AA(50),BB(50),CC(50),LEN(50)
      COMMON/CPLOT2/FTITLE,XPMIN,XPMAX,YPMIN,YPMAX,CHX,CHY
C        GET DMAX, DMIN
 100  TYPE 110
 110  FORMAT('$STARTING SGRID.  ENTER DMAX, DMIN : ')
      READ(5,*)DMAX,DMIN
      IF(DMAX.LT.DMIN)GOTO 100
C        INITIALIZE
      MS=0
      MCUM=0
      JCALL=0
C       ENTER X AT R=0
      TYPE 140
 140  FORMAT('$AT R=0, XO= : ')
      READ(5,*)XO
      XPLOT(1)=XO
      XLAST=XO
      X1=XO
C
C        ENTER ZONE DATA
      IZONE=0
 150  CONTINUE
      JCALL=JCALL+1
 160  TYPE 170,JCALL
 170  FORMAT(1X,' JCALL=',I2,/,3X,'ENTER TYPE(1=DECREASE,2=UNIFORM,'
     1 /,'$    3=INCREASE) OR 0=NO MORE GRIDS : ')
      READ(5,*)IT
      IF(IT.EQ.0)GOTO 300
      IF(IT.LT.1.OR.IT.GT.3)GOTO 160
      TYPE 190,XLAST
 190  FORMAT('$      X (FIRST)=',F8.3,'.  ENTER X AT END : ')
      READ(5,*)XT
C        SAVE INPUTS
      KSEG(JCALL)=IT
      XL(JCALL)=XT
      GOTO(200,220,270),IT
```

```
C          SEGMENT WITH DECREASING GRID SIZE
 200   IF(IZONE.GT.0)GOTO 210
C          FIRST ZONE
       XI=DMAX+XO
       CALL SEGS(DMAX,DMAX,X1,XI,MCUM,MS,XLAST)
 210   X1=XLAST
       DLAST=DMIN
       CALL SEGS(DMAX,DMIN,X1,XT,MCUM,MS,XLAST)
       GOTO 150
C
C          SEGMENT OF UNIFORM GRID SIZE
 220   IF(IZONE.GT.0)GOTO 260
 230   TYPE 240
 240   FORMAT('$       ENTER 1 IF DFIRST=DMAX,  2 IF DFIRST=DMIN :  ')
       READ(5,*)IFIRST
       IF(IFIRST.LT.1.OR.IFIRST.GT.2)GOTO 230
       IF(IFIRST.EQ.1)DLAST=DMAX
       IF(IFIRST.EQ.2)DLAST=DMIN
 260   X1=XLAST
       CALL SEGS(DLAST,DLAST,X1,XT,MCUM,MS,XLAST)
       GOTO 150
C
C          SEGMENT WITH INCREASING GRID SIZE
 270   IF(IZONE.GT.0)GOTO 280
C          FIRST ZONE
       XI=DMIN+XO
       CALL SEGS(DMIN,DMIN,X1,XI,MCUM,MS,XLAST)
       X1=XLAST
 280   IF(DMAX/DMIN.GT.5.0)GOTO 290
       DFIRST=DMIN
       DLAST=DMAX
       X1=XLAST
       CALL SEGS(DFIRST,DLAST,X1,XT,MCUM,MS,XLAST)
       GOTO 150
 290   DFIRST=DMIN
       DLAST=DMIN*2.
       X1=XLAST
C          INTERMEDIATE TARGET X
       XI=X1+.40*(XT-X1)
       CALL SEGS(DFIRST,DLAST,X1,XI,MCUM,MS,XLAST)
C          SECOND SEGMENT
       DFIRST=DLAST
       DLAST=DMAX
       X1=XLAST
       CALL SEGS(DFIRST,DLAST,X1,XT,MCUM,MS,XLAST)
       GOTO 150
C
 300   CONTINUE
       JCALL=JCALL-1
C          PRINT OUT DATA
       CALL PRINTX(XO,MS,MCUM,DMAX,DMIN,IFIRST,JCALL)
       STOP
       END
C
```

```
      SUBROUTINE SEGS(DFIRST, DLAST, X1, XT, MCUM, MS, XLAST)
C         JULY 1986
C         PURPOSE - TO FIND THE GRID STRETCHING PARAMETERS (A, B, C)
C                   FOR DECREASING GRID SIZE.  THE GRID STRETCHING IS
C                   X = A + B*R**C.       THE GIVENS ARE
C                   DFIRST, DLAST, X1, XT, MCUM.   THE OUTPUT IS MS,
C                   XLAST, XPLOT.
C         VARIABLES -
C               DFIRST = GRID SIZE AT X=X1
C                DLAST = GRID SIZE AT X=XLAST
C                  MCUM = NUMBER OF PRECEEDING R-GRIDS
C                    MS = NUMBER OF R-GRIDS IN THIS SEGMENT
C                    X1 = LOWER VALUE IN X-SPACE
C                XLAST = UPPER END VALUE IN X-SPACE
C                    XT = TARGET VALUE FOR UPPER END X VALUE
      COMMON/VALUES/XPLOT(200), JSEG(200), MMAX, KSEG(50), XL(361)
      COMMON/VALUE2/IZONE, IU, A1, B1, C1, AA(50), BB(50), CC(50), LEN(50)
C        INCREMENT THE ZONE NUMBER
      IZONE=IZONE+1
C        CHECK FOR UNIFORM GRIDS
      IF(DFIRST.EQ.DLAST)GOTO 150
C        ESTIMATE RE
      DEL=1.0E+10
      R1=MCUM
      ME=MAX1(3.,2.*(XT-X1)/SQRT(DFIRST*DLAST))
C        START LOOP HERE
      DO 120 M=2,ME
      R2=R1+FLOAT(M)
      C=1.+ALOG10(DLAST/DFIRST)/ALOG10(R2/R1)
      IF(C.EQ.0.0.OR.C.EQ.1.0)GOTO 120
C        UNBOUNDED VALUES
C     B=DFIRST/(C*(R1)**(C-1.))
C     A=X1-B*R1**C
C     X=A+B*R2**C
C        BOUNDED VALUES
      B=DFIRST/(C*EXP(AMIN1(80.,(C-1.0)*ALOG(R1))))
      A=X1-B*EXP(AMIN1(80.,C*ALOG(R1)))
      X=A+B*EXP(AMIN1(80.,C*ALOG(R2)))
      D=ABS(X-XT)
      IF(D.GE.DEL)GOTO 130
      DEL=D
      XLAST=X
      MS=M
      A1=A
      B1=B
      C1=C
  120 CONTINUE
  130 CONTINUE
      GOTO 200
C        UNIFORM GRIDS
  150 MS=AMAX1(1.,(XT-X1)/DFIRST+0.50)
      XLAST=X1+DFIRST*FLOAT(MS)
      B1=DFIRST
      A1=X1-B1*FLOAT(MCUM)
      C1=1.
      XLAST=X1+FLOAT(MS)*DFIRST
C        PRINT RESULT
  200 CALL WRTSAV(MCUM, MS, XLAST)
      MA=MCUM+1
      MB=MA+MS-1
      DO 220 M=MA,MB
      XPLOT(M)=A1+B1*FLOAT(M)**C1
      DX=XPLOT(M)-XPLOT(MAX0(1,M-1))
  220 CONTINUE
C        SAVE RESULTS
      MCUM=MCUM+MS
      LEN(IZONE)=MS
      RETURN
      END
```

```
      SUBROUTINE WRTSAV(MCUM,MS,X2)
      COMMON/VALUE2/IZONE,IU,A1,B1,C1,AA(50),BB(50),CC(50),LEN(50)
      AA(IZONE)=A1
      BB(IZONE)=B1
      CC(IZONE)=C1
      TYPE 100,MCUM,MS,X2
 100  FORMAT(8X,'MCUM=',I2,' MS=',I3,' X2=',F6.2)
      TYPE 110,A1,B1,C1
 110  FORMAT(8X,' A=',E9.3,' B=',E9.3,' C=',E9.3)
      R1=MCUM
      R2=MCUM+MS
      F1=A1+B1*R1**C1
      F2=A1+B1*R2**C1
      TYPE 120,F1,F2
 120  FORMAT(8X,' XFIRST,XLAST=',2F10.4)
      F1=C1*B1
      IF(C1-1.0.NE.0.0)F1=C1*B1*R1**(C1-1.)
      F2=C1*B1
      IF(C1-1.0.NE.0.0)F2=C1*B1*R2**(C1-1.)
      TYPE 130,F1,F2
 130  FORMAT(8X,' (DX/DR)FIRST,XLAST=',2F10.4)
      RETURN
      END
C
C-----------------------------------------------------------------------
C
      SUBROUTINE PRINTX(XO,MS,MCUM,DMAX,DMIN,IFIRST,JCALL)
C        PURPOSE - AT END OF RUN,  PRINT VALUES OF X, R TO SCREEN
      COMMON/VALUES/XPLOT(200),JSEG(200),MMAX,KSEG(50),XL(361)
      COMMON/VALUE2/IZONE,IU,A1,B1,C1,AA(50),BB(50),CC(50),LEN(50)
      COMMON/CPLOT2/FTITLE,XPMIN,XPMAX,YPMIN,YPMAX,CHX,CHY
C       PRINT VALUES OF R,X
      MMAX=MCUM
      M=0
      IU=3
      CALL FOPEN(IU)
      WRITE(IU,305)M,XO
      WRITE(5 ,305)M,XO
 305  FORMAT(5X,'R=',I3,' X=',F8.3)
      DO 310 M=1,MCUM
      DX=XPLOT(M)-XPLOT(MAXO(1,M-1))
      IF(M.EQ.1)DX=XPLOT(1)-XO
      WRITE(IU,320)M,XPLOT(M),DX
 310  TYPE 320,M,XPLOT(M),DX
 320  FORMAT(5X,'R=',I3,' X=',F8.3,'  D=',F8.3)
C        WRITE SEGEMENT INPUTS TO FILE
      WRITE(5,330)DMAX,DMIN,XO,JCALL,IFIRST
      WRITE(IU,330)DMAX,DMIN,XO,JCALL,IFIRST
 330  FORMAT(/,1X,'DMAX=',F8.2,' DMIN=',F8.2,'  XO=',F8.3,'  JCALL=',I2,
     1 '  IFIRST=',I2)
      WRITE(5,340)(KSEG(J),XL(J),J=1,JCALL)
      WRITE(IU,340)(KSEG(J),XL(J),J=1,JCALL)
 340  FORMAT(6(I4,F8.3))
      WRITE(IU,350)
      WRITE(5,350)
 350  FORMAT(1X)
C        WRITE A, B, C
 400  WRITE(5,410)IZONE,(LEN(I),I=1,IZONE)
      WRITE(IU,410)IZONE,(LEN(I),I=1,IZONE)
 410  FORMAT(1X,21I3)
      WRITE(5,420)(AA(I),BB(I),CC(I),I=1,IZONE)
      WRITE(IU,420)(AA(I),BB(I),CC(I),I=1,IZONE)
 420  FORMAT(3(1X,E12.6))
C     CALL  REGEN(IZONE,XE)
C     CALL APLOT
      RETURN
      END
```